Math 4530-1

Drawing surfaces with Maple

Here are some plotted surfaces. By re-entering the commands you can replot them. You can use your mouse to move them around. You can adjust various options in the plotting tool bar once you've clicked on the particular plot, regardless of how you specified (or didn't specify) them in the original plotting command. Particular important options to play with and understand are:

(1) Projection=constrained vs. unconstrained
(2) the various ways of exhibiting axes
(3) the various styles in which to draw your surface

At the end of your homework set you are asked to write procedures to plot several general types of surfaces, e.g. surfaces of revolution and graphs. As the book says, once you've written one procedure you can write thousands, and now is a good time to get over the hurdle of writing one, if you haven't yet done so.

1) A helicoid: (page 61) The two sets of coordinate curves on this surface are families of vertical helices and families of horizontal lines. For a parametric surface plot Maple automatically draws the coordinate curves

```
> restart:
  with(plots):  #to enter multiline (lists of) command
      #use shift-return.
Warning, the name changecoords has been redefined
```

```
> heli:=[v*cos(u),v*sin(u),u]:
> plot3d(heli,u=0..2*Pi,v=-3..3,axes=boxed,scaling=constrained,
  title='Helicoid');
```



2) A torus of revolution: (page 60).

```
> torus:=[(2+cos(u))*cos(v),(2+cos(u))*sin(v),sin(u)]:
> plot3d(torus,u=0..2*Pi,v=0..2*Pi,axes=boxed,title='Torus');
```




```
>
```

3) A Mobius strip (like page 68, but not exactly the same): You want to rotate a line segment around the z-axis, say, and at the same time rotate the line segment in the vertical plane through the z-axis, so that by the time you return the segment has been turned upside down. If you think about it you might come up with...

```
> theta:=u->u/2;  #how much to rotate the
      #line segment in the vertical plane, if the segment's been
      #rotated amount u about the z-axis  You can make more-rotated
      #strips by chosing different theta(u)'s.
```

$$\theta := u \to \frac{1}{2} u$$

```
> mobi:=[(2+v*cos(theta(u)))*cos(u),(2+v*cos(theta(u)))*sin(u),
        v*sin(theta(u))]:
> plot3d(mobi,u=0..2*Pi,v=-1.5..(1.5),axes=boxed,title='Mobius
  Strip');
```




```
> theta:=u->3*u/2:
  mobi:=[(3+v*cos(theta(u)))*cos(u),(3+v*cos(theta(u)))*sin(u),
         v*sin(theta(u))]:
  plot3d(mobi,u=0..2*Pi,v=-1.5..(1.5),axes=boxed,title='Twisted
  Strip');
```


Twisted Strip
1.5
1
0.5
0
–0.5
–1
–1.5
–2
0
2
4
–4
–2
0
2
4

>